System : $x' = k\,x,\ y' = m\,y$

$x' = -x,\ y' = -y$

In[2]:=
```
sys1aplot = StreamPlot[{-x, -y}, {x, -5, 5}, {y, -5, 5}]
```

Out[2]=



In[3]:=
```
sys1alines = Plot[{0, x, 2 x, -x, -2 x, x / 2, -x / 2, 100 x},
    {x, -5, 5}, PlotStyle → {{Thick, Black}}, PlotRange → {-5, 5}];
Show[sys1aplot, sys1alines]
```

Out[4]=



The critical point (0,0) is a **stable node** or **sink**.

$x' = -x, y' = -2y$

```
sys1bplot = StreamPlot[{-x, -2 y}, {x, -5, 5}, {y, -5, 5}]
```




```
sys1blines = Plot[{0, x^2, -x^2, x^2 / 5, -x^2 / 5, 100 x},
   {x, -5, 5}, PlotStyle → {{Thick, Black}}, PlotRange → {-5, 5}];
Show[sys1bplot, sys1blines]
```



The critical point (0,0) is also a stable node.
Both are actually **asymptotically stable**: solutions close enough to (0,0) converge to it.

$x' = x,\ y' = y$

In[6]:=
```
sys1cplot = StreamPlot[{x, y}, {x, -5, 5}, {y, -5, 5}]
```

Out[6]=



In[7]:=
```
Show[sys1cplot, sys1alines]
```

Out[7]=



The critical point (0,0) is an **unstable node** or **source**.

$x' = x$, $y' = 3\,y$

```
sys1dplot = StreamPlot[{x, 3 y}, {x, -5, 5}, {y, -5, 5}]
```



```
sys1dlines = Plot[{0, x^3/4, -x^3/4, x^3/25, -x^3/25, 100 x}, {x, -5, 5},
  PlotStyle → {{Thick, Black}}, PlotRange → {-5, 5}]; Show[sys1dplot, sys1dlines]
```



The critical point (0,0) is also an unstable node.

$x' = x,\ y' = -y$

```
sys1eplot = StreamPlot[{x, -y}, {x, -5, 5}, {y, -5, 5}, StreamPoints → Automatic]
```



```
sys1elines = Plot[{0, 1 / x, -1 / x, 3 / x, -3 / x, 7 / x, -7 / x, 11 / x, -11 / x, 100 x},
   {x, -5, 5}, PlotStyle → {{Thick, Black}}, PlotRange → {-5, 5}];
Show[sys1eplot, sys1elines]
```



The critical point (0,0) is a **saddle point**.

System: $x' = 2y + x - 1$, $y' = x^2 - 2x - 3$

```
sys2xnull = ContourPlot[{x == -1, x == 3}, {x, -5, 5}, {y, -5, 5}, ContourStyle → Blue];
sys2ynull = ContourPlot[{2 y + x - 1 == 0}, {x, -5, 5}, {y, -5, 5}, ContourStyle → Red];
sys2plot =
  StreamPlot[{2 y + x - 1, x^2 - 2 x - 3}, {x, -5, 5}, {y, -5, 5}, StreamPoints → Fine];
```

```
Show[sys2xnull, sys2ynull]
```



```
Show[sys2xnull, sys2ynull, sys2plot]
```



Critical points : **unstable spiral point** at $(-1, 1)$, saddle point at $(3, -1)$

System: $x' = 4\,y,\ y' = -x$

```
sys3plot = StreamPlot[{4 y, -x}, {x, -5, 5}, {y, -5, 5}]
```




```
sys3contour[c_] = x^2 + 4 y^2 == c;
sys3lines = ContourPlot[Evaluate[Map[sys3contour, {1, 9, 25, 49, 81}]],
  {x, -5, 5}, {y, -5, 5}, ContourStyle → {{Thick, Black}}];
Show[sys3plot, sys3lines]
```



The critical point (0,0) is a **center**.
It is stable, but not asymptotically stable: solutions nearby (0,0) stay close, but do not converge to it.

System: $x' = -y - x\left(1 - x^2 - y^2\right)^2$, $y' = x - y\left(1 - x^2 - y^2\right)^2$

```
sys4plot =
 StreamPlot[{-y - x (1 - (x^2 + y^2))^2, x - y (1 - (x^2 + y^2))^2}, {x, -2, 2}, {y, -2, 2}]
```




```
sys4lines =
  ContourPlot[x^2 + y^2 == 1, {x, -2, 2}, {y, -2, 2}, ContourStyle → {{Thick, Black}}];
Show[sys4plot, sys4lines]
```



Limit cycle on curve $x^2 + y^2 = 1$; **asymptotically stable spiral point** at (0,0).